

1. INTRODUCCIÓ I COMANDES BÀSIQUES

El programa que utilitzarem en les sessions de pràctiques és el MAPLE. Es tracta d'un manipulador algebraic o sistema de computació algebraica, és a dir, realitza les operacions de manera simbòlica sense aproximacions numèriques. Com veurem ben aviat en aquesta primera pràctica de contacte, a diferència de la calculadora, només obtindrem aproximacions numèriques amb decimals quan nosaltres el forcem a fer-ho.

Els següent enllaç d'internet us pot ser d'utilitat:

- www.maplesoft.com (en particular, si aneu a Student Help Center o Application Center).

En la versió 10 del Maple es poden utilitzar diferents entorns de treball. Aquestes pràctiques estan preparades per a realitzar-se amb l'entorn Classic Maple 10 (heu d'executar-lo a través del menú de Programes).

1.1. INICIALITZACIÓ

En una sessió de MAPLE hi ha línies per executar comandes que són les que comencen amb el símbol `>` en vermell. Les línies de comandes han d'acabar amb punt i coma (;) i s'executen immediatament quan premem la tecla **Return**. Tot seguit el resultat apareix just a sota en color blau. També mostra missatges d'error informatius si la línia de comandes no té la sintaxi correcta. Existeix una variant del punt i coma (;) que és acabar la comanda amb dos punts (:), aleshores s'executa normalment però no treu el resultat per pantalla. En una mateixa línia pot haver-hi més d'una comanda sempre i quan cadascuna d'elles vagi seguida del corresponent punt i coma (;) o dos punts (:), aleshores s'executen totes de cop al prémer la tecla **Return**.

Totes les línies de la sessió són operatives: és a dir, podem anar enrera i modificar línies anteriors de manera que al tornar-les a executar es modifica el resultat segons els nous paràmetres.

És important adonar-se i recordar que el programa diferencia les lletres majúscules de les minúscules.

Quan engegem el programa, no es carreguen tots els procediments i comandes de que el MAPLE disposa. Hi ha llibreries especialitzades que contenen comandes més avançades i específiques. En funció del que es vulgui fer servir, haurem de carregar abans la llibreria corresponent fent servir les instruccions `with()` o `readlib()`. Per exemple, la llibreria `plots` és útil per a dibuixar gràfics, la llibreria `linalg` per a qüestions relacionades amb àlgebra lineal, la llibreria `DEtools` per a qüestions relacionades amb equacions diferencials, etc.

Abans de començar una sessió de MAPLE és recomanable executar la comanda `restart;` que buida el contingut en memòria de totes les variables.

```
> restart;
```

També és recomanable d'executar-la cada cop que comencem un tema, exercici, ... nou per a no tenir conflictes amb variables ja definides.

1.2. ARITMÈTICA I FUNCIONS ELEMENTALS

La primera cosa que necessitem saber per poder fer els primers càlculs amb MAPLE és com realitzar les operacions aritmètiques bàsiques. Aquestes operacions aritmètiques es realitzen utilitzant els símbols estàndard que ja coneixeu `+`, `-`, `/`, `*` i `^`. Vegeu els següents exemples.

```
> 2+3;
> 3/2+5/3;
```

Observeu que els càlculs amb fraccions es realitzen sense convertir-los a decimals i queden resolts en forma algebraica i no decimal (per això diem que MAPLE és un manipulador algebraic).

```
> 2+(-2);
> 2*3/5+7*8;
> 45/89;
> 41^7;
```

En el següent bloc d'exemples veureu com s'introdueixen les funcions elementals com les trigonomètriques (sinus (**sin**), cosinus (**cos**), tangent (**tan**)), l'arrel quadrada (**sqrt**), l'exponencial (**exp**), el logaritme neperià (**ln**), el valor absolut (**abs**)...

```
> sin(Pi/2);
```

En aquest exemple anterior hem vist com treballar amb el número pi π (**Pi**) i en els següents, veureu com introduir el número e.

```
> ln(1);
> exp(1);
```

Recordeu que el número pi s'escriu Pi i que el número e s'introdueix amb la funció exponencial com exp(1) (no utilitzeu mai la lletra e, cal utilitzar la funció exponencial).

Uns exemples més...

```
> cos(Pi/2);
> tan(Pi/2);
> tan(Pi);
> abs(-3);
> sqrt(9);
```

També disposem de les funcions inverses de les trigonomètriques com **arcsin**, **arccos**, i **arctan**.

```
> arctan(sqrt(3));
> arccos(1);
```

A més de l'arrel quadrada, podem obtenir arrels n-èssimes amb la comanda **root[n]**. Per exemple, l'arrel tercera de 8 es calcula com

```
> root[3](8);
```

Un altre exemple d'ús de la comanda **root**,

```
> root[4](16);
```

Què passa al següent exemple?

```
> sqrt(-1);
> root[2](-1);
```

1.3. APROXIMACIONS DECIMALS

MAPLE treballa algebraicament. Si volem obtenir expressions decimals per aproximar valors d'expressions numèriques aleshores hem de fer servir la comanda `evalf`. Per defecte ens mostra una aproximació amb 10 xifres.

```
> evalf(45/89);  
> evalf(45/29);
```

Com veure-ho en els següents exemples, podem indicar-li el número de xifres que volem obtenir.

```
> evalf(45/89,26);  
> exp(1);  
> evalf(exp(1));  
> Pi;  
> evalf(Pi,20);
```

També podem fixar el número de xifres d'un cop per a tota la sessió donant un valor a la variable de sessió `Digits`. Feu `Digits:=30` i observeu què passa al fer diferents operacions.

```
> Digits:=30;  
> evalf(Pi);  
> evalf(sqrt(2));
```

1.4. VARIABLES

En sessions llargues de MAPLE o quan hem d'utilitzar moltes vegades una expressió, és molt útil poder utilitzar variables per a guardar resultats, fórmules, funcions,... L'assignació d'un valor a una variable es fa mitjançant l'operador `:=`. Si només posem el signe `=` el Maple no fa cap assignació a la variable.

```
> a=4;  
> a;
```

Utilitzant `:=` fem una assignació d'un valor a una variable i MAPLE recordarà quin és el seu valor durant tota la sessió. Si volem introduir un nou valor a la variable hem de fer una nova assignació.

```
> a:=3;  
> a;  
> a^2-1;  
> a:=2;  
> a;  
> a^2-1;
```

Per a "netejar" la variable de la memòria hem de fer

```
> a:='a';  
> a;
```

Recordeu que podem netejar TOTES les variables de cop amb la comanda `restart`.
Observeu la diferència entre el resultat de les línies següents,

```
> a:=3;
> b:=2:
```

Ara comprovem quin valor té la variable `b`.

```
> b;
```

El que ha passat és que la línia acabada amb dos punts (`:`) s'ha executat igualment però no ens ha mostrat el resultat per pantalla.

1.5. COM MANIPULAR EXPRESSIONS

Moltes vegades el resultat obtingut per una comanda no es troba en la forma més simplificada possible. En aquest apartat veurem tres comandes que ens ajudaran a manipular expressions.

- **simplify**: La comanda `simplify` realitza simplificacions en les expressions algebraiques. Com veurem també simplifica expressions trigonomètriques fent servir igualtats i fòrmules. Analitzeu els següents exemples.

```
> (2*x+4*x^2)/x;
> simplify((2*x+4*x^2)/x);
```

Un altre exemple. Noteu que podem referir-nos a l'últim càlcul realitzat durant la sessió mitjançant el símbol `%`.

```
> exp(y*ln(y^3));
> simplify(%);
> x/(x-1)+1/(x+1);
> simplify(%);
```

Anem a veure alguns exemples d'expressions trigonomètriques que es poden simplificar,

```
> cos(x)^2+sin(x)^2;
> simplify(%);
> M:=sin(3*t)-sin(5*t);
> simplify(M);
> M:=cos(x)^5+sin(x)^4+2*cos(x)^2-2*sin(x)^2-cos(2*x); simplify(M);
```

- **factor**: aquesta comanda intenta agrupar l'expressió en productes de factors senzills.

```
> factor(3*x^2-10*x-8);
> factor(x^2*y+2*x*y+y);
> M:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);factor(M);
```

Podeu explicar el resultat de la següent línia?

```
> factor(3*x^2-10*x-9);
```

Resulta que la comanda `factor` no treballa bé amb funcions que contenen radicals, només amb enters i fraccions. Intenteu trobar les arrels del polinomi amb la comanda `solve` que estudiarem més endavant.

```
> solve(3*x^2-10*x-9=0);
```

També podem simplificar expressions trigonomètriques,

```
> sin(x)^2-cos(x)^2;
> factor(%);
```

- **expand**: La comanda oposada és la comanda **expand**. Donada una expressió agrupada en factors, el que fa és realitzar totes les operacions possibles. I aplica fórmules trigonomètriques si és necessari.

```
> expand((x^2+x+3)^4*(x-1));
> expand((x+2)^2*(3*x-3)*(x+5)); factor(%);
> expand(sin(2*x));
> M:=x^(1/2)*(x^(3/2)+x^(-1/2)); expand(M);
```

- **eval**: utilitzarem la comanda **eval** quan vulguem substituir valors concrets a alguna de les variables que apareixen en una expressió.

```
> eval(3*x^2+8,x=4);
> W:=3*x^2+8;
> eval(W,x=4);
> eval(W,x=5+u);
> expand(%);
```

Si tenim una expressió que depèn d'una variable o paràmetre i volem avaluar-la per a un valor concret del paràmetre, utilitzarem la comanda **subs** on s'especifica el valor del paràmetre i l'expressió. Observeu-ne el funcionament en els següents exemples. En el primer exemple volem calcular el resultat de substituir el valor $x = 4$ a l'expressió polinòmica $4x^2 + 1$.

```
> subs(x=4,4*x^2+1);
> subs(t=-1,ln(t^3+5)/t^5);
```

1.6. RESOLDRE EQUACIONS

Un cop sabem definir expressions amb variables i avaluar-les en un valor concret, ens pot interessar solucionar equacions en funció d'una o més variables. Per això MAPLE disposa de la comanda **solve** en la qual hem d'indicar l'equació i el nom de la variable. Si només hi ha una variable, no cal indicar-li. Comencem amb un exemple senzillet.

```
> solve(x^2+x-1=0,x);
> solve(x^2+x-1=0);
> solve(x^2+x+1=0);
```

Què ha passat amb l'últim exemple?

Si no escrivim la igualtat i només escrivim una expressió, per defecte troba les solucions obtingudes al igualar la expressió a zero.

```
> solve(x^2+x-1);
```

També podem resoldre sistemes d'equacions. La única cosa que hem de fer és introduir la llista d'equacions i la llista de variables entre claudàtors.

```
> solve({x+y=1,x-y=1},{x,y});
```

1.7. FUNCIONS

MAPLE distingeix entre expressions i funcions. En les expressions tots els paràmetres i números que hi apareixen es tracten igual. Les funcions són expressions que es poden avaluar en uns paràmetres que són variables i que s'identifiquen de manera especial al definir la funció. Es defineixen mitjançant l'operador `->`. Vegeu els següents exemples.

```
> restart;
> f:=x->ln(x);
```

Aleshores f és una funció però $f(x)$ és una expressió!

```
> f(x);
```

Les següents assignacions NO són vàlides per definir funcions: $f:=\ln(x)$, o $f(x):=\ln(x)$.

L'avantatge de treballar amb funcions és que es poden substituir valors a les variables de manera molt senzilla. Vegeu com funciona amb els següents exemples.

```
> f(a);
> f(aa);
> f(1);
> f(2);
> evalf(f(2));
> f(z);
```

Un altre exemple.

```
> h:=x->exp(x^2)/x;
> h(2);
> h(a);
```

- **unapply**: Donada una expressió, també podem convertir-la en una funció amb la comanda **unapply** que té com a paràmetres l'expressió i la variable de la nova funció.

```
> j:=unapply(x^2*y-2,x);
> j(x);
> j(1);
```

1.8. GRÀFIQUES DE FUNCIONS

MAPLE dibuixa gràfiques de funcions d'una variable amb la comanda **plot**. Cal especificar la funció $f(x)$ i l'interval de la variable x que volem representar. L'interval de la variable y és opcional. Els intervals es representen indicant els extrems separats per dos punts `..`, per exemple $x=-2..3$ representa l'interval $[-2,3]$. Fixeu-vos que els dos eixos no tenen sempre la mateixa escala. Si volem representar més d'una funció en uns mateixos eixos de coordenades aleshores cal escriure les funcions entre claudàtors i separades per comes, per exemple, $[f(x),g(x),h(x)]$.

```
> plot(x^2-3*x+1,x=-1..4);
```

En el següent exemple retallem la gràfica anterior en l'eix vertical.

```
> plot(x^2-3*x+1,x=-1..4,y=-2..2);
> g:=x->ln(x^2)/x;
```

```
> plot(g(x),x=-2..2,y=-10..10);
```

En el següent exemple veurem com representar diverses funcions en uns mateixos eixos de coordenades.

```
> plot([sin(x),cos(x)],x=-2*Pi..2*Pi,y=-2*Pi..2*Pi);
```

Podem especificar els colors per a cada funció afegint una opció de color al final de la comanda. Els colors s'assignaran en el mateix ordre que el de les funcions.

```
> plot([sin(x),cos(x)],x=-2*Pi..2*Pi,y=-2*Pi..2*Pi, color=[blue,magenta]);
```

Els colors que podem fer servir són:

aquamarine, black, blue, navy, coral, cyan, brown, gold, green, gray, grey, khaki, magenta, maroon, orange, pink, plum, red, sienna, tan, turquoise, violet, wheat, white, yellow.

Dibuixem les següents funcions:

```
> plot([sin(x),3*sin(x),(1/2)*sin(x)], x=-2*Pi..2*Pi,y=-2*Pi..2*Pi,
color=[blue,red,green]);
```

```
> plot([cos(x),cos(4*x),cos((1/4)*x)], x=-4*Pi..4*Pi,y=-4*Pi..4*Pi,
color=[blue,red,green]);
```

```
> plot([sin(x),sin(x+Pi/2),sin(x+Pi)], x=-2*Pi..2*Pi,y=-2*Pi..2*Pi,
color=[blue,red,green]);
```

1.8.1. Gràfiques interactives: smartplot i interactive

La nova versió de MAPLE ofereix la possibilitat de construir gràfiques de manera interactiva. Realitzeu la següent prova. Escriviu la següent línia,

```
> sin(x);
```

Amb el ratolí heu de marcar el resultat que apareix en blau i apretar el botó dret del ratolí. Veureu que apareix un menú que ens permet aplicar diferents procediments a la funció seleccionada. Seleccioneu **Plots** i 2-D. Què obteniu?

Podeu modificar els valors dels paràmetres de la gràfica apretant altre cop el botó dret del ratolí. Per exemple aneu a les opcions **Axes** ->**Range**. Aquí podem modificar els rangs tant de l'eix de les X com el de les Y.

Ara aneu a la següent línia de comandes i executeu la següent línia

```
> cos(x);
```

Tot seguit seleccioneu el resultat amb el ratolí i, apretant el botó esquerre del ratolí, arrossegueu el resultat fins a sobre de la gràfica anterior. Què ha passat?

- **smartplot**: és una comanda que ens permet fer la gràfica d'una funció i modificar-ne els paràmetres a posteriori com hem fet a l'exemple anterior (apretant el botó dret del ratolí a sobre de la gràfica seleccionada)

```
> smartplot(cos(2*x));
```

- **interactive**: aquesta comanda executa una finestra auxiliar que ens permet triar el tipus de gràfica que volem realitzar. Es troba al paquet `plots`, i per això cal carregar-lo.

```
> with(plots);
> interactive(cos(x/2));
```

Si cliqueu al botó d'opcions veureu que es poden triar molts paràmetres de la gràfica.

1.8.2. Dibuixar punts, segments i fletxes

Per dibuixar punts i segments també podem fer servir una de les opcions de la comanda `plot`. Només cal donar les coordenades (entre claudàtors) dels punts en una llista entre claudàtors, i posar l'opció `style=point`. Si posem l'opció `style=line`, aleshores uneix els punts de la llista amb línies. Les opcions `color` i `style` permeten modificar l'aparença del resultat.

```
> plot([[2,3]],style=point);
> plot([[2,3],[-2,5]],x=-5..5,y=-6..6,style=point, color=blue,symbol=circle);
> plot([[2,3],[-2,5]],x=-5..5,y=-6..6,style=line, color=blue);
```

Per dibuixar les fletxes utilitzarem una comanda que es troba en un paquet específic de MAPLE. Es tracta de la comanda `arrow` en el paquet `plots`. Es diferencia de les comandes anteriors en que les coordenades de la fletxa s'escriuen entre els símbols `<` i `>`. Per carregar un paquet de comandes utilitzem la comanda `with`.

```
> with(plots);
> arrow(<1,2>);
```

1.8.3. Combinar diversos estils de gràfic en uns mateixos eixos: la comanda `display` del paquet `plots`

Si no heu carregat el paquet `plots`, ho podeu fer ara amb la comanda `with`.

```
> with(plots);
```

Amb la comanda `display` podem representar alhora diversos gràfics que guardarem en variables. Quan es guarda un gràfic en una variable, el millor és acabar la comanda amb dos punts per a no omplir la pantalla amb tota la informació que s'emmagatzema.

```
> dib1:=plot([-3*x+5,9-x^2],x=-3..5):
> dib2:=plot([[-1,8],[4,7]],style=point,color=blue,symbol=circle):
> display(dib1);
> display(dib2);
> display(dib1,dib2);
```

1.9. LA COMANDA CONVERT

Mireu a l'ajuda de MAPLE per entendre que fa la comanda `convert`. Veureu que és una comanda molt versàtil que permet canviar el tipus de variable entre molts formats. Nosaltres ens concentrarem en la seva utilitat per convertir d'unitats els angles de graus a radians i al revés.

Mireu el següents exemples,

```
> convert(9,binary);
> convert(1.24532,fraction);
> (x^3+x)/(x^2-1); convert(%,parfrac);
```


1.9.1. CANVI D'UNITATS: DE GRAUS A RADIANS

La comanda `convert` ens permet canviar de graus a radians i a l'inrevès. Per a les funcions trigonomètriques, els arguments es donen sempre en radians.

```
> g:=x->sin(x);  
> g(Pi);  
> g(Pi/3);
```

Per a transformar angles de radians a graus hem de fer

```
> convert(Pi,degrees);  
> convert(Pi/4,degrees);
```

Per a transformar angles de graus a radians hem de fer

```
> convert(180*degrees,radians);  
> convert(45*degrees,radians);
```

Busqueu a l'ajuda de MAPLE totes les possibilitats de la comanda `convert`.

1.10. LLISTES I SUCCESSIONS

Podem guardar informació en llistes de manera que sigui més accessible. En MAPLE les llistes es tanquen amb claudàtors (`[]`) i podem accedir a qualsevol dels seus elements indicant la seva posició a la llista. Mireu el següent exemple.

```
> llista:=[2,4,6,8,10];  
> llista[1];  
> llista[3];
```

Podem fer servir llistes per guardar, per exemple, les arrels d'un polinomi.

```
> arrels:=[solve(4-6*x+4*x^2-x^3)];  
> arrels[1];  
> arrels[2];
```

També podem crear successions de manera automàtica amb la comanda `seq`. Per exemple, si volem calcular les potències de 2, 2^n des de $n = 1$ fins a $n = 10$, podem usar la comanda `seq` indicant la fórmula i els valors de n com $n = 1..10$.

```
> seq(2^n,n=1..10);  
> potencies:=[seq(2^n,n=1..10)];
```

Si volem saber el resultat de 2^9 , ho podem mirar a la llista.

```
> potencies[9];
```

Per acabar, la comanda `map` serveix per aplicar una comanda a tots els elements d'una llista o una seqüència. Per exemple, si volem obtenir el logaritme neperià de la successió anterior hem de fer

```
> map(ln,potencies);  
> evalf(%);
```

1.11. EXERCICIS PER PRACTICAR

1. Calcula $\sqrt{\cos(\pi/3)} + \ln(3^6)$ amb 17 xifres decimals.
2. Factoritza el polinomi $x^3 + 3x^2 + 3x + 1$.
3. Dibuixa en uns mateixos eixos de coordenades les gràfiques de $f(x) = 3x^2 + 3x$ i $g(x) = x + 1$ de manera que s'observin els punts de tall. (Solucioneu l'equació $f(x) = g(x)$ i representeu els punts obtinguts).
4. Simplifiqueu l'expressió $\frac{4}{4+x^2} + \frac{1}{2+x}$.

2. NÚMEROS COMPLEXOS

En aquest capítol veurem com treballar amb números complexos amb MAPLE i repassarem les contruccions bàsiques.

2.1. ARITMÈTICA AMB NÚMEROS COMPLEXOS

Com que comencem una secció nova i no utilitzarem cap de les variables creades, netejarem la memòria de la sessió.

```
> restart;
```

MAPLE també permet realitzar tot tipus d'operacions aritmètiques amb números complexos de manera senzilla. El número complex i es representa per I .

```
> I^2;
> sqrt(-1);
```

- **Part real i part imaginària:** Per a obtenir la part real i la part imaginària d'un número complex utilitzem les comandes `Re` i `Im` com en el següent exemple.

```
> Re(2+I);
> Im(2+I);
```

- **El mòdul d'un número complex:** Per a trobar el mòdul d'un número complex utilitzarem la comanda `abs`:

```
> abs(2+I);
```

És correcte el resultat?

- **El conjugat:** I amb la comanda `conjugate` podem calcular el conjugat d'un número complex fent

```
> conjugate(2+I);
```

2.2. INTERPRETACIÓ GEOMÈTRICA

Els números complexos venen determinats per un parell de números reals que es poden representar en el pla. La part real a l'eix horitzontal i la part imaginària a l'eix vertical. Per exemple, els números complexos $2 + 3i$ i $4 - i$ corresponen als punts $(2, 3)$ i $(4, -1)$ i es presenten de la següent manera.

```
> z1:=2+3*I;
> z2:=4-I;
> with(plots):
> a1:=arrow(<Re(z1),Im(z1)>, shape=arrow, color=blue):
> a2:=arrow(<Re(z2),Im(z2)>, shape=arrow, color=red):
> display(a1,a2);
```

Aleshores algunes operacions com la suma de números complexos es poden interpretar geomètricament com la suma de vectors.

2.3. OPERACIONS AMB COMPLEXOS

Les operacions bàsiques amb números complexos són la suma, resta, multiplicació i divisió. Observeu que es realitzen de la mateixa manera a com ho feiem al principi de la pràctica amb els números naturals.

Considerem dos números complexos z_1 i z_2

```
> z1:=2+I;
> z2:=-1+5*I;
```

I comencem calculant la seva suma

```
> z1+z2;
```

Recordeu que els números complexos es poden representar al pla com a vectors on la coordenada x correspon a la part real i la coordenada y correspon a la part imaginària.

Fem-ho ara gràficament. Per això introduïrem una llibreria de comandes que és molt útil a l'hora de fer representacions gràfiques d'objectes geomètrics com, per exemple, vectors. És la llibreria `plots`. Si mireu a l'ajuda del MAPLE podeu veure totes les comandes que conté i què fan. Una llibreria es carrega a la sessió amb la comanda `with`

```
> with(plots);
```

Observeu que apareix una llista amb totes les noves comandes disponibles.

Ara dibuixarem els vectors donats per la part real i imaginària de z_1 i z_2 i el vector suma (els anomenarem z_{11} , z_{22} i z_3 , respectivament). La comanda que utilitzarem és la comanda `arrow` que té com a paràmetres els vectors que volem dibuixar i que, a més, podem definir una serie d'opcions per determinar la forma i el color de la fletxa.

```
> z11:=arrow(<2,1>,shape=arrow,color=blue):
> z22:=arrow(<-1,5>,shape=arrow,color=green):
> z3:=arrow(<1,6>,shape=arrow):
> display(z11,z22,z3);
```

De manera anàloga a com es calcula la suma, també es calcula la resta, multiplicació i divisió de números complexos.

```
> z1-z2;
> z1*z2;
> z1/z2;
```

Algunes vegades quan tenim expressions amb variables i números complexos, el MAPLE no les escriu com a tals, és a dir, amb la part real i la part imaginària agrupades. Per exemple

```
> (1+I)*(x+I);
```

Observem que passa al multiplicar complexos de mòdul (o norma) 1. Siguin $z = \frac{\sqrt{2}}{2}(1+i)$ i $w = i$. Anem a representar-los juntament amb el seu producte.

```
> restart;
> z:=sqrt(5)/5*(1+2*I); w:=I; y:=z*w;
> with(plots):
> a1:=arrow(<Re(z),Im(z)>,color=blue):
> a2:=arrow(<Re(w),Im(w)>,color=red):
> a3:=arrow(<Re(y),Im(y)>,color=yellow):
```

```
> display(a1,a2,a3);
```

Fixeu-vos que el vector resultant torna a ser de mòdul 1 i el seu angle és la suma dels angles dels vectors que hem multiplicat.

- **Simplificar amb la comanda `evalc`**: La comanda `evalc` es fa servir per a expressions amb valors complexos i separa aquestes expressions en part real i part imaginària. Sempre que és possible el resultat de fer `evalc` es posa en la forma canònica `expr1 + I*expr2`.

```
> evalc((1+I)*(x+I));
```

Un exemple potser més interessant és el següent. Recordeu que els números complexos es poden expressar en la forma d'Euler. Observeu que ens dóna quan li demanem que ens expressi e^{ix} en forma de número complex.

```
> z:=exp(I*x);
> Re(z);
> Im(z);
> evalc(z);
```

Així doncs, es fàcil adonar-se que MAPLE també ens permetrà calcular de manera molt senzilla les potències i exponencials amb números complexos.

```
> z:=2-3*I;
> exp(z);
> evalc(%);
> evalf(%,20);
```

Observeu en l'exemple anterior com hem utilitzat la comanda `evalf` per a obtenir una aproximació decimal del resultat amb 20 xifres.

2.4. FUNCIONS I RESOLUCIÓ D' EQUACIONS AMB NÚMEROS COMPLEXOS

Tot el que hem fet referent a funcions i expressions amb números reals es pot fer amb números complexos. És a dir, podem solucionar equacions utilitzant la comanda `solve`. Mireu que passa quan resollem l'equació $x^2 + x + 1 = 0$.

```
> solve(x^2+x+1=0);
```

Aquesta equació no té solucions reals com molt bé sabeu (o podeu comprovar) i aleshores MAPLE ens dóna les solucions complexes directament. Vegeu un exemple en que la pròpia equació conté coeficients que són números complexos,

```
> solve(x^2-I*x+4*I=0);
```

Les equacions polinòmiques de graus superiors no sempre són tan senzilles de resoldre i aleshores MAPLE utilitza mètodes d'aproximació numèrica per trobar algunes solucions.

```
> solve(x^5+4*x^3-2*x^2+1=0);
> evalf(%);
```

Anem a resoldre l'equació $z^3 = 1$ i a representar gràficament les solucions. Recordeu que per dibuixar vectors ja hem vist que podem utilitzar la comanda `arrow` que es troba a la llibreria `plots`.

```
> solve(z^3=1);
> with(plots):
> arrow([<1,0>,<-1/2,1/2*sqrt(3)>,<-1/2,-1/2*sqrt(3)>]);
```

Anem a fer un exemple una mica més complicat creant llistes (recordeu el que vam fer a la pràctica anterior). Intenteu deduir que fa cadascuna de les següents línies de codi.

```
> solve(z^3=I);
> evalf(solve(z^3=I));
> solucions:=[evalf(solve(z^3=I))];
> coordx:=map(Re,solucions);
> coordy:=map(Im,solucions);
> vectors:=seq(<coordx[i],coordy[i]>, i=1..3);
> punts:=seq([coordx[i],coordy[i]], i=1..3);
> with(plots):
> arrow([vectors]);
> plot([punts],style=point);
```

De la mateixa manera es poden definir funcions amb números complexos, avaluar, etc però no podem utilitzar la comanda plot per dibuixar-les ja que no es poden representar al pla real.

```
> f:=x->x^2*I-exp(x);
> f(I);
> evalc(%);
> f(Pi);
> evalc(%);
> evalf(%);
```

2.5. EXERCICIS PER PRACTICAR

1. Calculeu:

$$(3+2i)(3-2i), |3+2i|, \frac{3-2i}{-1+4i}, \frac{(3-2i)(-1-4i)}{(-1+4i)(-1-4i)}, (2+2i)^{12}$$

2. Comproveu la igualtat $|z_1 \cdot z_2| = |z_1| \cdot |z_2|$ on $z_1 = 3+2i$ i $z_2 = -1-4i$. És cert que $|z_1 + z_2| = |z_1| + |z_2|$? Doneu un exemple.
3. Trobeu totes les solucions de $z^5 = 1$ i dibuixeu-les. Feu el mateix per $z^5 = i$.
4. Comproveu que $2+i$ és una arrel del polinomi $p(x) = x^2 - 4x + 5$ (recordeu que un número a és una arrel d'una funció $f(x)$ si es compleix que $f(a) = 0$).
5. Expressen $\sqrt{1+I}\sqrt{3}$ de la forma $a+bI$, on a i b són reals. Doneu el resultat tant en forma algebraica com mitjançant una aproximació decimal.
6. Feu una gràfica on es vegin les funcions $\sin(x)$, $\sin(\frac{1}{2}x)$ i $\sin(2x)$. Quina diferència observeu? Quina és la periodicitat de cadascuna d'elles?

2.6. FORMES POLARS I CARTESIANA

Per tal de passar de forma polar a cartesiana, i a l'inrevés, el MAPLE fa servir, bàsicament, la comanda `polar(z)`. En general, per a obtenir l'expressió cartesiana cal forçar les instruccions amb la comanda `evalc(z)`.

Així per a trobar la forma polar del complex $1 + \sqrt{3}i$ escriurem

```
> polar(1+sqrt(3)*I)
```

per a obtenir $2e^{\frac{\pi}{3}i}$ escrit donant el mòdul i l'argument per separat,

$$\text{polar}(2, \frac{1}{3}\pi)$$

2.7. EXERCICIS PER PRACTICAR

1. Troba la forma polar i/o cartesiana dels complexos següents

$$3e^{5\pi i}, \sqrt{2}e^{\frac{3}{2}\pi i}, 1+4i, i, e^{2,5\pi}, (1+\sqrt{3}i)(1-\sqrt{2}i)$$

2. Dóna una forma polar equivalent amb l'angle entre 0 i 2π (és a dir, que tinguin la mateixa forma cartesiana) pels complexos següents

$$3e^{\frac{21}{4}\pi i}, \sqrt{2}e^{101\pi i}, e^{-\frac{235}{3}\pi i}, e^{100i}$$

3. Donats els números complexos $z_1 = 1+i$, $z_2 = 1-2i$, $z_3 = 2e^{\frac{7\pi}{3}i}$, $z_4 = 3e^{\frac{9\pi}{4}i}$, calcula

$$z_1^3, z_1(z_2-1)^4, z_3^4 z_4^2, |z_1+z_2|, \frac{1-\bar{z}_1}{z_2+1}, z_2 z_1^4 + z_3^2 \frac{\bar{z}_1+z_2}{z_1-z_2}, ||z_1|+z_2|$$

4. Donats els números complexos $z = 1+2i$, comprova que s'obté el mateix resultat, calculant $(-2+2i)^{24}$ directament o buscant primer la forma polar i fent servir les propietats del producte de complexos en forma polar.

2.8. ARRELS COMPLEXES

A més de la instrucció bàsica `^` hi ha una altra comanda que ens permet escriure potències fraccionàries d'una expressió, aquesta és `root`. Així les línies següents són equivalents:

```
> (-2+2*I)^(1/3);
> root[3](-2+2*I);
```

Aquest mètode de trobar arrels d'un número complex només troba la de l'argument més petit. Per a forçar l'obtenció de les tres arrels de $-2+2i$, cal escriure

```
> solve(x^3-(-2+2*I)=0);
```

i s'obté

$$(1+I)(-\frac{1}{2} + \frac{1}{2}I\sqrt{3}), (1+I)(-\frac{1}{2} - \frac{1}{2}I\sqrt{3}), 1+I$$

Per a obtenir l'expressió desenvolupada cal usar la comanda `evalc`, però com s'ha d'aplicar a un objecte de tres elements separats per comes, primer s'ha de convertir a una llista (per això afegim els símbols `{ }`), i després usar la comanda `map` per tal d'aplicar la funció `evalc` a tota la llista,

```
> map(evalc, {solve(x^3+2-2*I=0,x)});
```

per obtenir

$$\{1+I, -\frac{1}{2} - \frac{1}{2}\sqrt{3} + I(-\frac{1}{2} + \frac{1}{2}\sqrt{3}), -\frac{1}{2} + \frac{1}{2}\sqrt{3} + I(-\frac{1}{2} - \frac{1}{2}\sqrt{3})\}$$

Per veure totes les arrels obtingudes en forma gràfica es pot usar el programa `dibuixarrels` que es troba a continuació. Per rotar el conjunt de les arrels es pot usar la instrucció `rotate` (cal el paquet `plottools`).

El que ve a continuació és una rutina o procediment feta amb Maple. No entrarem en detall de com programar en MAPLE però el que heu de fer és copiar les següents línies de codi sense preocupar-vos del missatge d'avís Warning que us anirà sortint mentre escriviu.

```
> dibuixarrels:=proc(A)
> local llista,i;
> llista=[];
> for i from 1 to nops(A) do
> llista:=[op(llista),[Re(op(i,A)),Im(op(i,A))]];
> od;
> llista:=[op(llista),[Re(op(1,A)),Im(op(1,A))]];
> plot(llista,symbol=circle,style=point,scaling=constrained);
> end;
```

Així per veure, gràficament, les tres arrels anteriors només cal escriure

```
> dibuixarrels(map(evalc,{solve(x^3+2-2*I=0,x)}));
```

Usant la comanda

```
> polar(1+I);
```

podem saber quin angle hem de girar per veure el multiple de les arrels de la unitat corresponents a aquest exemple.

```
rotate(dibuixarrels(map(evalc,{solve(x^3+2-2*I=0,x)})), -Pi/4)
```

Podem veure els dos triangles equilaters en el mateix gràfic usant la comanda `display` (cal el paquet `plots`)

```
> with(plots):with(plottools):
> a:=dibuixarrels(map(evalc,{solve(x^3-1=0,x)})):
> b:=rotate(dibuixarrels(map(evalc,{solve(x^3+2-2*I=0,x)})), -Pi/4):
> display(a,b);
```

2.9. Exercicis

1. Troba totes les arrels complexes dels números següents,

$$\sqrt[8]{-2}, \sqrt[5]{125i}, \sqrt[3]{-1-i}, \sqrt[4]{2}$$

2. Comprova, girant l'angle adient, que totes les arrels complexes dels números següents estan sobre els radis d'un pentagon regular.

$$\sqrt[5]{1}, \sqrt[5]{2}, \sqrt[5]{i}$$

2.10. SOLUCIONAR UNA EQUACIÓ

Recorda que la instrucció per resoldre una equació es el `solve`. En el cas que es volen trobar les arrels d'un polinomi es pot usar la funció `factor` per descomposar-lo en producte d'altres de grau més petit. La funció `evalf` es pot usar per escriure les expressions numèriques de les arrels.

2.11. EXERCICIS

1. Comprova usant la fórmula per resoldre una equació de segon grau i directament usant el `solve` del MAPLE que s'obté el mateix resultat al resoldre les equacions següents,

$$z^2 = i, z^2 + z + 2 = 0$$

2. Soluciona les equacions de grau superior següents (usa la funció `subs` per comprovar que tot el que ha trobat el MAPLE son solucions)

$$z^3 + z^2 - 4z - 4 = 0, z^3 + z^2 + 2z + 1 = 0, z^3 + (i - 1)z^2 + (1 - i)z - 1 = 0, z^{10} - z + 1 = 0$$

2.12. GRÀFIQUES AL PLA COMPLEX

Per tal de dibuixar en coordenades polars el MAPLE fa servir la opció `coords=polar` en la funció `plot`. Primer definirem la funció del radi r depenent de l'angle t , $r(t)$, i aleshores escriurem la següent comanda `plot([r(t),t,t=0..2*Pi],coords=polar)`.

Així $(r(t), t)$ és la corba en polars que es vol descriure.

Si volem dibuixar una circumferència de radi 1 amb centre a l'origen, $x^2 + y^2 = 1$, que en coordenades polars és $r = 1$, la instrucció en MAPLE per obtenir aquesta gràfica és

```
> plot([1,t,t=0..2*Pi],coords=polar,scaling=constrained);
```

Nota: La opció `scaling=constrained` s'usa per garantir que l'escala vertical i l'horitzontal del dibuix siguin iguals.

2.13. EXERCICIS

1. Dibuixa la circumferència de radi 2 i centre $(0, 0)$.
2. Per dibuixar la corba $r(\theta) = 1 + \theta$ cal fer-ho en coordenades polars. Comprova que s'obtenen resultats diferents si es fan servir les dues instruccions següents:

```
> plot(1+t,t,t=0..2*Pi);
> plot([1+t,t,t=0..2*Pi],coords=polar);
```

2.14. FUNCIONS PERIÒDIQUES: FASORS

Una ona elemental de freqüència ω és una funció $f(t) = A \sin(\omega t + \varphi)$, on A és l'amplitud, i φ és la fase. El seu període fonamental és $T = \frac{2\pi}{\omega}$. Les funcions que s'obtenen per superposició d'ones elementals (fent sumes i restes) són especialment importants. Aquestes ones es poden expressar com $B \cos(\omega t) + C \sin(\omega t)$ on $B = A \sin \varphi$ i $C = A \cos \varphi$.

- Fem una representació gràfica de $3 \sin(2t + \pi) + 4 \cos(2t + \frac{\pi}{3})$. El resultat obtingut correspon a una funció periòdica? En cas afirmatiu, quin és el seu període fonamental i la seva freqüència?
- ```
> plot(3*sin(2*t+Pi)+4*cos(2*t+Pi/3),t=0..6*Pi);
```

En general, la suma d'ones elementals de la mateixa freqüència torna a ser una ona elemental de la mateixa freqüència.

Si sumem ones elementals de diferents freqüències, el resultat obtingut no té perquè ser una ona elemental, i, de fet, ni tan sols té perquè ser una funció periòdica.

- Feu una representació gràfica de  $3 \sin(\sqrt{2}t + \pi) + 4 \cos(2t + \frac{\pi}{3})$ . El resultat obtingut correspon a una funció periòdica?
- Feu una representació gràfica de  $3 \sin(2t + \pi) + 4 \cos(3t + \frac{\pi}{3})$ . El resultat obtingut correspon a una funció periòdica?

En general, la suma d'ones elementals  $A_1 \sin(\omega_1 t + \varphi_1) + A_2 \sin(\omega_2 t + \varphi_2)$  és periòdica si i només si  $\frac{\omega_1}{\omega_2} = \frac{p}{q}$  és un número racional (és a dir, un quocient de números enters). Aleshores si  $\frac{p}{q}$  és una fracció irreduïble, tenim que el període fonamental de la suma és  $\frac{2\pi p}{\omega_1}$ .

En els dos últims exemples veiem que el primer cas no correspon a una funció periòdica (ja que  $\frac{\omega_1}{\omega_2} = \frac{\sqrt{2}}{2}$  no és un número racional), i el segon exemple sí que correspon a una funció periòdica de període fonamental  $2\pi$  (ja que  $\frac{\omega_1}{\omega_2} = \frac{2}{3}$  que és un número racional expressat en forma irreduïble, i el període fonamental és  $\frac{2\pi p}{\omega_1} = \frac{2\pi 2}{2} = 2\pi$ ).

## 2.15. EXERCICIS

Expresseu les següents funcions com a suma d'ones elementals, feu-ne una representació gràfica, digueu si la funció és periòdica o no, i en cas que ho sigui calculeu el període fonamental:

1.  $f(t) = \sqrt{2} \sin 2t + \cos(2t)$
2.  $f(t) = 2 \sin(3t + \frac{\pi}{3}) + \sin(\sqrt{5}t + \pi)$
3.  $f(t) = \sin(5t) + \sin(15t + \frac{\pi}{5})$